

Mathematics 244: Lab 5

Fall 2003

0. Introduction and Setup. In this lab, we shall use Maple to study the qualitative properties of autonomous systems of two differential equations.

Please turn in only the printout of your Maple worksheet. Include explicit answers to all questions asked, using the **text** feature of Maple to insert them in the worksheet. Remove from the worksheet any extraneous material and any errors you have made.

Before you start up Maple, copy the seed file into your directory from the Web page of the course. As in Lab4, we require the `DEtools`, `plots`, and `LinearAlgebra` packages. There are also some constants that will be used in both of the remaining sections. To make these definitions, the seed file contains the following instructions.

```
restart;  
with(plots): with(DEtools): with(LinearAlgebra):  
window:=x=-6..6,y=-6..6;  
trange:=-6..6;
```

Each numbered section of this lab deals with a system of two **autonomous** differential equations, i.e., a system of the form

$$x'(t) = F(x, y), \quad y'(t) = G(x, y).$$

Note that the distinguishing feature of an autonomous system is that the functions F and G are independent of the variable t . This allows many properties of the solutions to be studied using the curves, called **trajectories**, that show the path in the xy plane followed by the solutions (It is an easy exercise to show that if an initial condition is on a trajectory, then the whole solution follows that trajectory). The Maple command `DEplot` may be used to draw trajectories and direction fields for such systems.

1. A linear system.

Consider the linear system: $x' = x + 2y$, $y' = -5x - 2y$. You teach Maple to recognize this using the instructions

```
delx:= diff(x(t),t) = x(t) + 2*y(t);  
dely:= diff(y(t),t) = -5*x(t) - 2*y(t);
```

that appear in the seed file for lab5. This system has $(0, 0)$ as an equilibrium point. We will study the stability of that point.

1a. The Direction field. Use the following instruction to recall previous definitions and construct a plot of the direction field of this system. The plot consists of small arrows pointing the way of the trajectories in the square $-6 \leq x \leq 6$, $-6 \leq y \leq 6$.

```
df1:=DEplot([delx,dely],[x(t),y(t)], trange, window):
```

(note the colon at the end to suppress output of the plot structure). You can show the plot using the line `df1;` (with a semicolon this time).

1b. Nullclines. The points where the direction field is **horizontal** (characterized by $dy/dt = 0$) or **vertical** (characterized by $dx/dt = 0$) form curves called **nullclines**. In many cases, these curves provide useful information about the behavior of trajectories without the excessive detail of a direction field. They can be plotted by the following instructions (the color is intended to distinguish these curves on the screen — it is not necessary to print your report in color).

```
dh1:=plot(-5*x/2,window,color=coral):
dv1:=plot(-x/2,window,color=violet):
display({df1,dh1,dv1});
```

Note that the nullclines usually cut across the arrows in the slope field since they are **not** solutions of the equation (except in rare cases). The line `dh1`, colored `coral` shows where the direction field is horizontal, and the line `dv1`, colored `violet` shows where the direction field is vertical.

1c. Trajectories and stability. To study the stability of an equilibrium point, it is also useful to draw the direction field together with several trajectories. To draw the direction field together with the trajectories through the points $[-1, 2]$, $[.5, .7]$, $[0.1, -1]$, and $[-2, -3]$, execute the commands

```
inits:={ [x(0)=-1,y(0)=2], [x(0)=0.5,y(0)=0.7], [x(0)=0.1,y(0)=-1],
[x(0)=-2,y(0)=-3] };
DEplot([delx,dely],[x(t),y(t)], trange, inits, window);
DEplot([delx,dely],[x(t),y(t)], trange, inits, window, stepsize=.1);
```

Note the difference between these two plots. In the first one, the trajectories do not follow the arrows very closely, while in the second one, they do. When `DEplot` is used, the plot is generated from approximations to the solution at points at a distance `stepsize` apart. If this number is too large, the plot will be inaccurate. The use of the option `stepsize` allows us to change the default value (in this case `.6`) to a smaller value (in this case `.1`) to obtain a more accurate plot.

To draw the trajectories without the arrows, execute the command

```
DEplot([delx,dely],[x(t),y(t)], trange, inits,
window, stepsize=0.1, arrows=NONE);
```

1d. Classification. Based on your plots, classify the critical point at the origin and determine whether it is stable or not.

1e. Eigenvalues. Use the `Eigenvalues` command to calculate the eigenvalues of the matrix of the system of part (a) and **discuss** the relation of this result to the graph in part (c) and the classification in part (d).

2. A nonlinear system.

Consider the almost linear system

$$x' = 2y - 2x + xy - x^2 = (2+x)(y-x)$$

$$y' = 4y + 4x - xy - x^2 = (4-x)(y+x).$$

The equilibrium solutions are $[x = 0, y = 0]$, $[x = -2, y = 2]$, and $[x = 4, y = 4]$. Maple can obtain these by using

```
F2:=2*y - 2*x + x*y - x^2;
G2:=4*y + 4*x - x*y - x^2;
eqpts:=solve({F2,G2},{x,y});
```

2a. The Direction field. Use the `DEplot` command to obtain a plot of the direction field of this system using the values of `trange` and `window` defined earlier.

2b. Nullclines. In this example, the factored form of the expressions for dx/dt and dy/dt allows a simple parametric description of the nullclines. In particular, the slope field is horizontal along `dh2`, whose plot can be constructed using

```
dh2:=plot([[4,t,t=trange],[t,-t,t=trange]],window,color=coral):
```

Construct a similar instruction to produce a plot showing where the slope field is vertical (you should use `color=violet` as before), and obtain a `display` combining the slope field and both sets of nullclines. The equilibrium points should be seen as points lying on one nullcline of each color.

2c. Zooming in. Isolate each equilibrium point. That is, for each equilibrium solution (x_0, y_0) , use `DEplot` to plot the direction field in the rectangle $x_0 - 1 \leq x \leq x_0 + 1$, $y_0 - 1 \leq y \leq y_0 + 1$. Add some typical trajectories to your plots which illustrate the behavior near each equilibrium solution (make sure the `stepsize` option is chosen properly). Using these plots, classify each equilibrium point state and state whether it appears to be stable or unstable.

2d. Eigenvalues. The type and stability of the critical points can also be determined by examining the eigenvalues of the corresponding linear system. The entries of matrix of the linear system corresponding to each critical point can be obtained by Maple by executing the following sequence of commands, which calculate the partial derivatives of the right hand sides of the differential equations, assembles them into a matrix, and then substitutes the coordinates of the critical point for x and y .

```
A11:= diff(F2,x);
A12:= diff(F2,y);
A21:= diff(G2,x);
A22:= diff(G2,y);
A:= < <A11 | A12>, <A21 | A22> >;
Aa:= subs(eqpts[1],A);
Eigenvectors(Aa);
```

You should find the other **linearizations** in a similar fashion using the names `Ab` and `Ac` for the matrices.

2e. Discussion. Describe the relation between the graphs found in part (c) and the algebraic results in part (d).